

Module 3

Basic FOCUS Commands

In this module you will learn how to:

- Begin and end requests
- Display data
- Sort report columns
- Select records

Supporting Files/Documents:

- FOCUS Documentation
- CIRS Data Element Dictionary
- Active Current Status Master File Description

FOCUS Report Requests

FOCUS is a non-procedural reporting language. It uses simple English-like phrases to generate ad hoc reports. The basic type of FOCUS report is a tabular report format where the data is arranged in columns.

This workbook will provide only the most basic commands to generate this type of report. Refer to your FOCUS documentation for a complete list of commands and reporting options.

The basic components of a tabular report request, in recommended order, are:

- Execution of standard defines
- Beginning the report request
- Specifying display fields
- Sorting report data
- Selecting records
- Ending the report request

Beginning and Ending Report Requests

The FOCUS reporting language is very flexible and only requires a command to identify the data source and a command to mark the end of the request. All other parts of the request are optional. Although not required, we recommend including comments in your request and executing the standard defines for the file you are using. The syntax is:

```
-*text  
EX filename  
TABLE FILE filename  
(optional commands)  
END
```

-* text	Comments you provide to identify the purpose of the request. Each comment line must begin with a hyphen and an asterisk (-*).
EX filename	Creates fields identified with an asterisk (*) in the Master File Description. Filename is any file containing defined fields. Defines must be executed before the TABLE FILE command.
TABLE FILE filename	Initializes the request and specifies the data source for the report request. Filename is any FOCUS database, external file or extracted file.
(optional commands)	All of the other parts of the request are optional - you only need to include the commands and phrases that produce the report you want. This module will discuss the optional commands used to display data, sort report columns and select records.
END	Ends the report request and generates the data. This command must be typed on a line by itself.

Displaying Data

Basic reports retrieve field values from a database and displays those values. The four display commands - also known as verbs - are PRINT, LIST, SUM and COUNT. The syntax of a display command is:

verb fieldname [AND] fieldname

verb	Is the PRINT, LIST, SUM or COUNT command.
fieldname	Is the name of the field to be displayed. It is also known as the display field. A maximum of 256 display fields can be requested in a report. The fields will appear on the report in the same order in which they are specified. For defined fields, the column title on the report will be the fieldname; otherwise the column title will be a 'reader friendly' title.
AND	Is an optional keyword used to improve the readability of the report request. It can be used between any fieldnames and does not affect the report.

Display Command: PRINT

The display command PRINT will display the values of the field(s) specified in your report request.

Report Request

```
-* LIST OF EMPLOYEES  
EX AC  
TABLE FILE AC  
PRINT AC:SSA AC:WNAME  
END
```

Report Generated

```
AC:SSA          AC:WNAME  
-----          -  
001-11-0000    BEAR, JAMES  
002-22-0000    RATTLESNAKE, JOHN P  
003-33-0000    SEAL, DONALD      F  
004-44-0000    GIRAFFE, JOAN  
005-55-0000    WHALE, BETSY G
```

Display Command: LIST

LIST works the same as PRINT, except it also numbers the values in the report. If LIST is used in a report request that includes a sort phrase, the list counter will reset to 1 each time the value in the first sort field changes.

Report Request

```
-* LIST OF EMPLOYEES
EX AC
TABLE FILE AC
LIST AC:SSA AC:WNAME
END
```

Report Generated

```
LIST  AC:SSA          AC:WNAME
----  -
1      001-11-0000    BEAR, JAMES
2      002-22-0000    RATTLESNAKE, JOHN P
3      003-33-0000    SEAL, DONALD      F
4      004-44-0000    GIRAFFE, JOAN
5      005-55-0000    WHALE, BETSY G
```

Display Command: SUM

The display command SUM will add all values of the field you specify for the records in the database and display one summary line. SUM, WRITE and ADD are synonyms and can be used interchangeably. If SUM is used with alphanumeric or date fields, SUM will not add the values; instead, it will display the last value retrieved from the database.

Report Request

```
-* TOTAL OF EMPLOYEE SALARIES  
EX AC  
TABLE FILE AC  
SUM AC: SALARY  
END
```

Report Generated

```
AC: SALARY  
-----  
$37,698.93
```

Display Command: COUNT

The display command COUNT will count the number of occurrences of the field in the database. By default, a COUNT field has an I5 usage format.

Report Request

```
-* NUMBER OF SALARY FIELDS  
EX AC  
TABLE FILE AC  
COUNT AC : SALARY  
END
```

Report Generated

```
AC : SALARY  
COUNT  
-----  
          6
```


Sorting Report Columns

Sorting enables you to group report data in a particular order. Data can be sorted in a column or across a row. The syntax of a sort command is:

sort fieldname

sort Is the BY or ACROSS command.

fieldname Is the name of the sort field. Any field in the database can be a sort field.

- Only 1 fieldname can follow a sort command, but multiple sort phrases can be used in a report request.
- Values for the sort field are displayed only once in the report.
- By default, the sorting sequence is from low to high. The sorting sequence can be reversed by using HIGHEST. For example:

BY HIGHEST AC:SALARY

- The sequence of sort phrases determines the sorting order. Each successive sort is nested within the previous sort.
- You can create a matrix by including both rows and columns in the same report.

Sort Command: BY

The sort command BY is used to sort data into rows.

- Up to 32 BY sort phrases can be used in a report.
- When used with the display command LIST, the counter is reset to 1 each time the major (first) sort value changes.

Report Request

```
-* EMPLOYEE LIST
-* BY SEX
EX AC
TABLE FILE AC
PRINT AC:FRSTNAME
BY AC:SEX
BY AC:LASTNAME
END
```

Report Generated

SEX	AC:LASTNAME	AC:FRSTNAME
---	-----	-----
F	BEAR	ANDREA
	JAGUAR	AMY A
	SEAL	JOAN M
M	HIPPO	MANUAL
	MONKEY	PAUL

Sort Command: ACROSS

The sort command ACROSS is used to sort data into columns.

- Up to 5 ACROSS sort phrases can be used in a report.
- Each ACROSS phrase can retrieve up to 95 sort field values.
- The total number of ACROSS report columns is equal to the total number of ACROSS sort field values multiplied by the total number of display fields. The maximum number of columns is 256.

Report Request

```
-* EMPLOYEE COUNTS
-* ACROSS SEX
EX AC
TABLE FILE AC
COUNT AC:SSA
ACROSS AC:SEX
END
```

Report Generated

```
SEX
  F      M
-----
  3      2
```

Selecting Records

You can include selection (screening) criteria in your report requests using the WHERE command. The syntax is:

IF / WHERE fieldname operator literal

IF / WHERE Is the command phrase for selecting records. The command IF offers a subset of WHERE functionality.

fieldname Is the field you want to test (the test value).

operator Is the type of selection operator.

value Is the alphanumeric or numeric value for the corresponding fieldname. Note: Blank values and values containing spaces must be enclosed in single quotation marks.

- You can use as many IF / WHERE expressions as necessary to define your selection criteria. All criteria must be satisfied in order for a record to be included in a report.
- Simple expressions can be joined with AND and OR logical operators to specify complex selection criteria.

Record Limits

You can limit the number of records FOCUS retrieves that meet your selection criteria. The syntax is:

IF / WHERE RECORDLIMIT EQ number

- Number is any number greater than 0, and indicates the number of records to be retrieved.
- Limit records when you are developing a new report and only need a few records to verify the results and/or design.
- Include a record limit to reduce costs if you know the number of records that meet the test criteria, so that the search does not continue beyond the last record retrieved.

Types Of Selection Tests

Below are examples of operators used in IF / WHERE expressions.
Additional operators can be found in your FOCUS User Manual.

Code	Description	Example
	Equal	WHERE AC:SEX EQ 'F'
	Not Equal	WHERE AC:SEPCODE NE ' '
	Greater Than	WHERE AC:SALARY GT 3000
	Less Than	WHERE AC:FTE LT 1.0
	Greater Than or Equal	WHERE AC:APPTXDTE GE 07012001
	Less Than or Equal	WHERE AC:EMPDATE LE 06301998
	Range	WHERE AC:AGE FROM 50 TO 65
	Not in Range	WHERE AC:UNIT NOT-FROM '300' TO '500'
	Include character string	WHERE AC:WNAME CONTAINS 'JONES'
	Arithmetic Operations	WHERE AC:SALARY/173.33 GT 25.00

Example of a screening statement using IF:

```
IF AC:EMPCBID EQ R02 OR R05 OR R07 OR R09
```

Example of a screening statement using WHERE:

```
WHERE AC:EMPCBID EQ 'R02' OR 'R05' OR 'R07' OR 'R09'
```

Example of a compound screening statement – only used with WHERE:

```
WHERE (AC:EMPCBID EQ 'R03')  
      OR (AC:CLASS EQ '2353' OR '2354' OR '2363')
```

Summary Of Basic Commands

Function	Syntax	Example
Comment line	-* text	-* ACTIVE EE LIST
Execute standard defines	EX filename	EX AC
Begin report request	TABLE FILE filename	TABLE FILE AC
Display command (verb phrase)	verb field(s)	PRINT AC:WNAME AC:CLASS
Sort phrase	sort field	BY AC:AGENCY BY AC:UNIT
Select records	Screening statement(s)	IF AC:SEPCODE EQ ` `
Limit the records retrieved	recordlimit	IF RECORDLIMIT EQ 5
End report request	END	END

Practice

1. Write a report that displays the class code, time base and salary of your active and on-leave employees. Sort the report by employee name.

Report Generated:

AC:WNAME	CLS	AC:FTE	AC:SALARY
-----	---	-----	-----
BEAR, CLARK F	2358	.200	\$1,275.00
FROG, NANCY B	3312	1.000	\$6,346.00
LLAMA, JOAN M	2360	.500	\$3,492.00
	2360	.500	\$3,343.50
MONKEY, RON J	1038	1.000	\$2,917.00

2. Write a report to generate the total number of active and on-leave positions at your campus. Sort by CBID.

Report Generated:

AC:EMPCBID	AC:POSITION COUNT
-----	-----
C99	3
E99	11
M80	50
R03	334

Module 3

3. Identify the names and job codes of your active and on-leave Management Personnel Plan employees. Sort the report by employment date.

Report Generated:

AC:EMPDATE -----	AC:WNAME -----	JOB CODE -----
01/03/1997	KANGAROO, MARY E	M113
08/24/1979	CAMEL, JAMES J	M092
09/01/1983	ADVARK, DIANNE T	L025
08/07/1989	NEWT, GRACE P	N062
01/08/1990	WHALE, PETER A	Z244

4. Identify the total salary for your monthly paid active and on-leave employees. Sort the report data by reporting unit and by class code.

Report Generated:

UNT ---	CLS ---	AC:SALARY -----
110	1170	\$3,664.00
	3306	\$7,865.00
610	0400	\$27,259.00
	0420	\$4,695.00
	3300	\$13,371.00

5. Write a report to count your active and on-leave Faculty positions.
Sort the report by ethnic group and across sex.

Report Generated:

	SEX	
	F	M
AC:ETHNICGRP		

AMI	1	0
ASN	17	17
BLK	11	16
HSP	10	8
N/A	0	1
WHT	211	208

What You Have Learned

In this module you have learned to:

- Begin and end requests
- Display data using verb phrases
- Sort the data in your reports
- Select records